SMD is a simple absolute dumper which runs entirely within the onboard monitor RAM from C6AH to D9CH. Its starting address is C6A hex. When run, it clears the screen and expects an encoding specification and filename just as the 4.Ø resident loader. After these are input, the starting and ending hex addresses are input as shown in the folloiwng example where the SMD is used to copy itself:

Key in → SPJCGA (CR) G
(Screen Cleared, Cursor in Upper Left)

B
SMD (CR)
C6A,D9D(CR)* (D9D used for safety)
C6A
D6A
D6A (This last is an endrecord)
(Screen clears again, ready for another dump)

*Before data is dumped, the cassette recorder should be setup with the proper plug in the microphone jack. The Byte/Biphase cassette card has two plugs for writing - one for byte and one for biphase. The read plug (labelled usually "EAR" or "SPKR") should not be plugged in. Also make sure that enough tape runs before typing the final carriage return on the end address specification so that non-recordable leader gets a chance to pass by before dumping starts.

The onboard dumper was hand optimized to fit inside the free space on system RAM, but the system stack also resides there. This means that the stack may over-run the dumper, erasing part of it. If the dumper has been in RAM while BASIC has been run, for example, the stack has probably squashed it at some time. If there is doubt, check the byte at D9CH. It should be a C9 (return instruction). If it is not, or you just want to make sure, reload the dumper just before using it.

When the dumper is dumping, each record will be displayed as a hex number on the screen. The hex number represents the address of the data being dumped on each record. That address is put on the header of the record so the 4.Ø resident loader will know where to put it when it is read back in.

The last record is an "END" type record. It is put on automatically. It will display as a record with dump address equal to the address of the record before it. Optimization of the dumper's code requires some strangeness such as this, but in any case, the last record (dump finished) will be signaled by the screen clearing. This puts the dumper back in its initial mode, just as if it had been restarted at C6AH. More data may be dumped if desired.

```
                        ; ******* ONBOARD DUMPER FOR 4.0 *******
                        ;
                        ;       THIS IS A POLYFORMAT DUMPER FOR ABSOLUTE
                        ;DATA WHICH RUNS FROM C6A TO D9F (OR SO), START ADDRESS
                        ;C6AH.  WHEN RUN, IT ACTS LIKE 4.0 MONITOR TAPE LOAD IN
                        ;THE WAY IT ACCEPTS ENCODING SPECIFICATION (B OR P) AND
                        ;FILE NAME.  THEN IT EXPECTS TWO HEX NUMBERS FOR
                        ;START AND END DUMP ADDRESSES.  EACH RECORD DUMPED SHOWS
                        ;ADDRESS USED IN HEX ON SCREEN.  WHEN DONE, IT PUTS OUT
                        ;AN "END" TYPE RECORD AND CLEARS SCREEN, READY
                        ;FOR ANOTHER DUMP.
                        ;
                        ;ORIGINAL 2.2 DUMPER SYSTEM WRITTEN BY DAVID FAIMAN
                        ;REWRITTEN,DOCUMENTED AND CONVERTED TO ONBOARD FOR 4.0
                        ;BY R.L.DERAN
                        ;
                        ;
0C20                    WH0     EQU     0C20H
0C24                    WH1     EQU     0C24H
0C16                    SRA4    EQU     0C16H
02AD                    SETUP   EQU     02ADH
03AA                    HEXC    EQU     03AAH
03D1                    DEOUT   EQU     03D1H
0C5A                            ORG     0C5CH-2
0C5A                    LENGTH: DS      2
0C5C                    WNAME:  DS      8
0C64                    WRN:    DS      2
0C66                    WLEN:   DS      1
0C67                    WADR:   DS      2
0C69                    WTYPE:  DS      1
0C6A 21450D             START:  LXI     H,TISR
0C6D 22160C                     SHLD    SRA4
0C70 3E0C               STAR2:  MVI     A,0CH   ;FORM FEED
0C72 CD240C                     CALL    WH1     ;CLEAR SCREEN
0C75 CD200C                     CALL    WH0
0C78 CD240C                     CALL    WH1
0C7B FE42                       CPI     'B'
0C7D CA920C                     JZ      BITE
0C80 FE50                       CPI     'P'
0C82 C2700C                     JNZ     STAR2
0C85 CDAD02             POLY:   CALL    SETUP
0C88 05                         DB      005H
0C89 AA                         DB      0AAH
0C8A 40                         DB      040H
0C8B 0C                         DB      00CH
0C8C E6                         DB      0E6H
0C8D E6                         DB      0E6H
0C8E 00                         DB      000H
0C8F C39A0C                     JMP     NAMER
0C92 CDAD02             BITE:   CALL    SETUP
0C95 06                         DB      006H
0C96 AA                         DB      0AAH
0C97 40                         DB      040H
```

```
0C98 CE                 DB      0CEH
0C99 00                 DB      000H
                ;
                ;       NAMEING ROUTINE
                ;
0C9A 210000     NAMER:  LXI     H,0
0C9D 22640C             SHLD    WRN
0CA0 0E08               MVI     C,8     ;BLANK NAME FIELD
0CA2 21630C             LXI     H,WNAME+7
0CA5 3620       NAM:    MVI     M,020H
0CA7 2B                 DCX     H       ;BACKUP H TO WNAME
0CA8 0D                 DCR     C
0CA9 C2A50C             JNZ     NAM
0CAC 23                 INX     H
0CAD 0E08               MVI     C,8
0CAF CD180D             CALL    CRLF
0CB2 CD200C     NAM0:   CALL    WH0
0CB5 CD240C             CALL    WH1
0CB8 FE0D               CPI     00DH    ;CR
0CBA CAC30C             JZ      DUMPC
0CBD 77                 MOV     M,A
0CBE 23                 INX     H
0CBF 0D                 DCR     C
0CC0 C2B20C             JNZ     NAM0
0CC3 AF         DUMPC:  XRA     A
0CC4 32690C             STA     WTYPE
0CC7 CD180D             CALL    CRLF
0CCA CDAA03     SIZE:   CALL    HEXC
0CCD 22670C             SHLD    WADR
0CD0 78                 MOV     A,B
0CD1 CD240C             CALL    WH1
0CD4 EB                 XCHG
0CD5 CDAA03             CALL    HEXC
0CD8 CD180D             CALL    CRLF
0CDB 7D                 MOV     A,L
0CDC 93                 SUB     E
0CDD 6F                 MOV     L,A
0CDE 7C                 MOV     A,H
0CDF 9A                 SBB     D
0CE0 67                 MOV     H,A
0CE1 225A0C             SHLD    LENGTH
0CE4 CDF60C             CALL    DUMPR
0CE7 3E02       ENDC:   MVI     A,2
0CE9 32690C             STA     WTYPE
0CEC 3D                 DCR     A
0CED 32660C             STA     WLEN
0CF0 CD540D             CALL    DUMP
0CF3 C36A0C             JMP     START
                ;
                ;       DUMP DATA RECORDS
                ;
0CF6 215B0C     DUMPR:  LXI     H,LENGTH+1
0CF9 7E                 MOV     A,M
0CFA B7                 ORA     A
0CFB CA100D             JZ      OVER
0CFE 35                 DCR     M
0CFF AF                 XRA     A
0D00 32660C             STA     WLEN
```

```
0D03 CD540D          CALL    DUMP
0D06 2A670C          LHLD    WADR
0D09 24              INR     H
0D0A 22670C          SHLD    WADR
0D0D C3F60C          JMP     DUMPR
0D10 2B      OVER:   DCX     H
0D11 7E              MOV     A,M
0D12 32660C          STA     WLEN
0D15 C3540D          JMP     DUMP
                ;
0D18 3E0D    CRLF:   MVI     A,0DH
0D1A CD240C          CALL    WH1
0D1D C9              RET
                ;
                ;       ROUTINE TO OUTPUT A RECORD
                ;
0D1E 0600    PUT:    MVI     B,0     ;CLEAR CHECKSUM
0D20 4F              MOV     C,A     ;PUT LENGTH OF RECORD IN C
0D21 7E      PUT0:   MOV     A,M
0D22 23              INX     H
0D23 F5              PUSH    PSW
0D24 80              ADD     B
0D25 47              MOV     B,A
0D26 F1              POP     PSW
0D27 CD340D          CALL    TO
0D2A 0D              DCR     C
0D2B C2210D          JNZ     PUT0
0D2E 78              MOV     A,B
0D2F 2F              CMA
0D30 3C              INR     A
0D31 C3340D          JMP     TO
                ;
                ;       TAPE OUTPUT ROUTIME
                ;
0C08         TBUFF   EQU     0C08H
0D34 E5      TO:     PUSH    H
0D35 21080C          LXI     H,TBUFF
0D38 F5              PUSH    PSW
0D39 7E      TO1:    MOV     A,M
0D3A B7              ORA     A
0D3B C2390D          JNZ     TO1
0D3E 23              INX     H
0D3F F1              POP     PSW
0D40 77              MOV     M,A
0D41 2B              DCX     H
0D42 34              INR     M
0D43 E1              POP     H
0D44 C9              RET
                ;
                ;       TISR IS A SIMPLE USART READER WHICH WILL
                ;       RE-TRANSMIT THE CHARACTER IN TBUFF IF IT HAS NOT
                ;       BEEN REPLACED BY THE WORMHOLE ROUTINE.  IT
                ;       DOES NOT CHECK THE FLAG, BECAUSE IT ASSUMES
                ;       THAT THE PROGRAM CALLING THE WORMHOLE IS FASTER
                ;       THAN THE USART AND SO IT ALWAYS HAS A VALID
                ;       CHARACTER FOR US TO TAKE.
                ;
0D45 AF      TISR:   XRA     A
```

```
0D46 32080C            STA     TBUFF
0D49 3A090C            LDA     TBUFF+1
0D4C D300              OUT     0
0D4E E1       IORET:   POP     H
0D4F D1                POP     D
0D50 C1                POP     B
0D51 F1                POP     PSW
0D52 FB                EI
0D53 C9                RET
                ;
                ;       DUMP PUTS OUT ONE COMPLETE RECORD.
                ;       IT TURNS ON USART AND MOTORS,WAITS A WHILE
                ;       FOR AN IRG, PUTS OUT 64 SYNCH CHARACTERS,
                ;       DUMPS A RECORD ACCORDING TO THE WRITE CONTROL
                ;       BLOCK  AT WNAME (IT ALSO PUTS THE WCB
                ;       ON THE RECORD AS HEADER), INCREMENTS THE RECORD
                ;       NUMBER, STOPS USART AND MOTORS, AND RETURNS.
                ;
0D54 3E21     DUMP:    MVI     A,021H
0D56 D301              OUT     1
0D58 2A670C            LHLD    WADR
0D5B EB                XCHG
0D5C CDD103            CALL    DEOUT   ;DISPLAY THE ADDRESS WE'RE DUMPI
0D5F CD180D            CALL    CRLF
0D62 21FF8F            LXI     H,08FFFH
0D65 2B       DELAY:   DCX     H
0D66 7C                MOV     A,H
0D67 B7                ORA     A
0D68 C2650D            JNZ     DELAY
0D6B 0E40              MVI     C,64
0D6D 3EE6              MVI     A,0E6H  ;SYNC CHARACTER
0D6F CD340D   DUMP0:   CALL    TO
0D72 0D                DCR     C
0D73 C26F0D            JNZ     DUMP0
0D76 3E01              MVI     A,001H  ;START OF HEADER
0D78 CD340D            CALL    TO
                ;
                ;       DUMP HEADER AND DATA RECORDS
                ;
0D7B 3E0E              MVI     A,00EH  ;LENGTH OF HEADER RECORD
0D7D 215C0C            LXI     H,WNAME
0D80 CD1E0D            CALL    PUT
0D83 3A660C            LDA     WLEN
0D86 2A670C            LHLD    WADR
0D89 CD1E0D            CALL    PUT
0D8C 21640C            LXI     H,WRN
0D8F 34                INR     M
0D90 AF       OFF:     XRA     A
0D91 CD340D            CALL    TO      ;THESE PUSH OUT LAST BYTES FROM
0D94 CD340D            CALL    TO      ;THE USART AND WH BUFFER PIPELIN
0D97 CD340D            CALL    TO      ;TURN OFF MOTOR AND TRANSMITTER
0D9A D301              OUT     1
0D9C C9                RET
0000                   END
```